O_h irreps and decompositions

In[1]:=
```
OhRed = <|
  "Irreps" → {"A1g", "A2g", "Eg", "T1g", "T2g"},
  "CharacterTable" →
   {
    {1, 1, 1, 1, 1},
    {1, 1, -1, -1, 1},
    {2, -1, 0, 0, 2},
    {3, 0, -1, 1, -1},
    {3, 0, 1, -1, -1}
   },
  "ConjugacyClassSizes" → {1, 8, 6, 6, 3}
  |>;
```

In[2]:=
```
Oh = <|
  "Irreps" → {"A1g", "A2g", "Eg", "T1g", "T2g", "A1u", "A2u", "Eu", "T1u", "T2u"},
  "CharacterTable" →
   Flatten[KroneckerProduct[
     (1  1
      1 -1),
     {
      {1, 1, 1, 1, 1},
      {1, 1, -1, -1, 1},
      {2, -1, 0, 0, 2},
      {3, 0, -1, 1, -1},
      {3, 0, 1, -1, -1}
     }], {1}],
  "ConjugacyClassSizes" → {1, 8, 6, 6, 3, 1, 8, 6, 6, 3}
  |>;
```

In[3]:=
```
IsotypicDecomposition[G_Association, char__] := Module[{multi},
  multi = Table[1/Total[G["ConjugacyClassSizes"]]
      (G["ConjugacyClassSizes"] × G["CharacterTable"][[i]]).char,
     {i, Length[G["CharacterTable"]]}];

  multi = DeleteCases[multi G["Irreps"], 0];

  If[Length[multi] > 1, multi /. List → CirclePlus, multi[[1]]]
 ]
```

```
In[4]:= χO3[l_, θ_] := ChebyshevU[2 l, Cos[θ / 2]] (*:= Sin[(l + 1/2) θ] / Sin[1/2 θ] *)
        χO3Red[l_] := χO3[l, #] & /@ {0, 2 π / 3, π, π / 2, π}
        χO3Full[l_] := Flatten[KroneckerProduct[{1, (-1)^l}, χO3Red[l]], 1]
        χ[name_, G_ : Oh] := G["CharacterTable"][[Position[G["Irreps"], name][[1, 1]]]]

In[8]:= l = 1;
        χO3Red[l]
        χO3Full[l]
        IsotypicDecomposition[Oh, χO3Full[l]]

Out[9]= {3, 0, -1, 1, -1}

Out[10]= {3, 0, -1, 1, -1, -3, 0, 1, -1, 1}
```

Out[11]= $T_{1 u}$

```
In[12]:= l = 2;
         χO3Red[l]
         χO3Full[l]
         IsotypicDecomposition[Oh, χO3Full[l]]

Out[13]= {5, -1, 1, -1, 1}

Out[14]= {5, -1, 1, -1, 1, 5, -1, 1, -1, 1}
```

Out[15]= $E_g \oplus T_{2 g}$

```
In[16]:= l = 3;
         χO3Red[l]
         χO3Full[l]
         IsotypicDecomposition[Oh, χO3Full[l]]

Out[17]= {7, 1, -1, -1, -1}

Out[18]= {7, 1, -1, -1, -1, -7, -1, 1, 1, 1}
```

Out[19]= $A_{2 u} \oplus T_{1 u} \oplus T_{2 u}$

```
In[20]:= l = 4;
         χO3Red[l]
         χO3Full[l]
         IsotypicDecomposition[Oh, χO3Full[l]]

Out[21]= {9, 0, 1, 1, 1}

Out[22]= {9, 0, 1, 1, 1, 9, 0, 1, 1, 1}
```

Out[23]= $A_{1 g} \oplus E_g \oplus T_{1 g} \oplus T_{2 g}$

Superconductivity Order Parameter

In[24]:=
```
D4 = <|
   "Irreps" → {"A1", "A2", "B1", "B2", "E"},
   "CharacterTable" →
    {
     {1, 1, 1, 1, 1},
     {1, 1, 1, -1, -1},
     {1, -1, 1, 1, -1},
     {1, -1, 1, -1, 1},
     {2, 0, -2, 0, 0}
    },
   "ConjugacyClasses" → {"E", "C4", "C2z", "C2'", "C2''"},
   "ConjugacyClassSizes" → {1, 2, 1, 2, 2}
   |>;
```

In[25]:= (*Matrix rep in the basis $\{e^{ik_x}, e^{ik_y}, e^{-ik_x}, e^{-ik_y}\}$*)

NNBasis = $\{e^{i k_x}, e^{i k_y}, e^{-i k_x}, e^{-i k_y}\}$;

D4gens = {

(*C_4*) $\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{pmatrix}$,

(*$\sigma_x = C_2'$ *) $\begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$

};

```
MatrixForm /@
 (D4mats = Flatten[Table[MatrixPower[D4gens[[1]], i].MatrixPower[D4gens[[2]], j],
     {i, 0, 3}, {j, 0, 1}], 1])
```

Out[27]=

$$\left\{\begin{pmatrix} 1&0&0&0\\0&1&0&0\\0&0&1&0\\0&0&0&1 \end{pmatrix}, \begin{pmatrix} 0&0&1&0\\0&1&0&0\\1&0&0&0\\0&0&0&1 \end{pmatrix}, \begin{pmatrix} 0&1&0&0\\0&0&1&0\\0&0&0&1\\1&0&0&0 \end{pmatrix}, \begin{pmatrix} 0&1&0&0\\1&0&0&0\\0&0&0&1\\0&0&1&0 \end{pmatrix}, \begin{pmatrix} 0&0&1&0\\0&0&0&1\\1&0&0&0\\0&1&0&0 \end{pmatrix}, \begin{pmatrix} 1&0&0&0\\0&0&0&1\\0&0&1&0\\0&1&0&0 \end{pmatrix}, \begin{pmatrix} 0&0&0&1\\1&0&0&0\\0&1&0&0\\0&0&1&0 \end{pmatrix}, \begin{pmatrix} 0&0&0&1\\0&0&1&0\\0&1&0&0\\1&0&0&0 \end{pmatrix}\right\}$$

In[28]:= D4matClasses = {"E", "C2'", "C4", "C2''", "C2z", "C2'", "C4", "C2''"};

In[29]:=
```
projectorD4[irrep_] := Module[{repind},
  repind = Position[D4["Irreps"], irrep][[1, 1]];
  D4["CharacterTable"][[repind, 1]]/8 Sum[D4["CharacterTable"][[repind,
     Position[D4["ConjugacyClasses"], D4matClasses[[i]]][[1, 1]]]] × D4mats[[i]], {i, 1, 8}]
 ]
```

In[30]:=
```
projectors = {PA1, PA2, PB1, PB2, PE} = projectorD4 /@ D4["Irreps"];
```

In[31]:=
```
TableForm[{MatrixForm /@ projectors}, TableHeadings → {None, D4["Irreps"]}]
```

Out[31]//TableForm=

A_1	A_2	B_1	B_2	E
$\begin{pmatrix} \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}$	$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$	$\begin{pmatrix} \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} \end{pmatrix}$	$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$	$\begin{pmatrix} \frac{1}{2} & 0 & -\frac{1}{2} & 0 \\ 0 & \frac{1}{2} & 0 & -\frac{1}{2} \\ -\frac{1}{2} & 0 & \frac{1}{2} & 0 \\ 0 & -\frac{1}{2} & 0 & \frac{1}{2} \end{pmatrix}$

In[32]:=
```
#.# == # & /@ projectors
```

Out[32]= {True, True, True, True, True}

In[33]:=
```
Eigensystem[PA1]
```

Out[33]= {{1, 0, 0, 0}, {{1, 1, 1, 1}, {-1, 0, 0, 1}, {-1, 0, 1, 0}, {-1, 1, 0, 0}}}

In[34]:=
```
Normalize[{1, 1, 1, 1}].NNBasis // ExpToTrig
```

Out[34]= $\mathrm{Cos}[k_x] + \mathrm{Cos}[k_y]$

In[35]:=
```
Eigensystem[PB1]
```

Out[35]= {{1, 0, 0, 0}, {{-1, 1, -1, 1}, {1, 0, 0, 1}, {-1, 0, 1, 0}, {1, 1, 0, 0}}}

In[36]:=
```
Normalize[-{-1, 1, -1, 1}].NNBasis // ExpToTrig
```

Out[36]= $\mathrm{Cos}[k_x] - \mathrm{Cos}[k_y]$

In[37]:=
```
Eigensystem[PE]
```

Out[37]= {{1, 1, 0, 0}, {{0, -1, 0, 1}, {-1, 0, 1, 0}, {0, 1, 0, 1}, {1, 0, 1, 0}}}

In[38]:=
```
Normalize[{0, -1, 0, 1}].NNBasis // ExpToTrig
Normalize[{-1, 0, 1, 0}].NNBasis // ExpToTrig
```

Out[38]= $-i\sqrt{2}\,\mathrm{Sin}[k_y]$

Out[39]= $-i\sqrt{2}\,\mathrm{Sin}[k_x]$

In[40]:=
```
DensityPlot[Cos[kx] + Cos[ky], {kx, -π, π}, {ky, -π, π},
 MeshFunctions → {#3 &, #3 &}, Mesh → {{0}, {0}}, MeshStyle → Opacity[0.5, Black]]
```

Out[40]=



In[41]:=
```
DensityPlot[Cos[kx] - Cos[ky], {kx, -π, π}, {ky, -π, π},
 MeshFunctions → {#3 &, #3 &}, Mesh → {{0}, {0}}, MeshStyle → Opacity[0.5, Black]]
```

Out[41]=



In[42]:=
```
DensityPlot[Sin[kx], {kx, -π, π}, {ky, -π, π}, MeshFunctions → {#3 &, #3 &},
  Mesh → {{0}, {0}}, MeshStyle → Opacity[0.5, Black]]
```

Out[42]=



In[43]:=
```
DensityPlot[Sin[kx] Sin[ky] (Cos[kx] - Cos[ky]), {kx, -π, π},
  {ky, -π, π}, MeshFunctions → {#3 &, #3 &}, Mesh → {{0}, {0}},
  MeshStyle → Opacity[0.5, Black], PlotPoints → 150]
```

Out[43]=